

Objectifs résolution d'équations différentielles, du premier et du deuxième ordre, linéaires ou non linéaires, en utilisant la commande "ode" de Scilab.

Compte-rendu : exercice E2, questions 4, 5, 6.

1. ÉQUATIONS DIFFÉRENTIELLES D'ORDRE 1

1.1. L'appel de la commande ode de Scilab

a) Résolution numérique d'une équation différentielle d'ordre 1

Soit une équation différentielle pour la fonction $y(t)$ de la forme suivante (voir TP info[2] §1.1b) : $y'(t) = f(t, y(t))$

Conditions aux limites : définie par $(t_0, y_0) \in \mathbb{R}^2$, abscisse t_0 et ordonnée y_0 telle que $y_0 = y(t_0)$.

On cherche la solution du problème (la fonction $y : t \mapsto y(t)$) par un algorithme d'approximation numérique. La fonction ode de Scilab utilise des méthodes sophistiquées que nous ne chercherons pas à expliquer.

S'agissant d'une solution numérique, rappelons que nous n'obtiendrons pas l'expression littérale de la fonction $y(t)$, mais plutôt une série de valeurs de y à partir d'une série de valeurs de t .

b) les arguments de ode

L'appel de la fonction ode nécessite l'entrée dans l'ordre de quatre arguments :

- la valeur initiale y_0 (a priori $y_0 = y(0)$).
- l'abscisse initiale t_0 (a priori $t_0 = 0$).
- les valeurs de t pour lesquelles on veut calculer la solution approchée.
- la fonction y_{prime} traduisant l'équation différentielle à résoudre.

Avec tout cela :

```
y = ode(y0,t0,t,yprime)
```

affecte à y la liste des valeurs de la fonction solution approchée de l'équation différentielle pour les valeurs de t données.

c) Exemple très simple

```
y0 = 1
t0 = 0
t = linspace(0,10,100); // "t =" est facultatif
function yprime = yprime(t,y)
    yprime = -y
endfunction
y = ode(y0,t0,t,yprime)
clf()
plot2d(t,y,style=5)
```

Exercice E1

Quelle est l'équation différentielle résolue par le script ci-dessus ?

Quelle en est la solution théorique ?

Tracer le graphe de cette solution théorique, et comparer au tracé donné par le script. Suggestion : tracer les deux graphes sur la même figure, avec deux couleurs différentes, après avoir décalé l'une des deux fonctions de 10^{-3} , puis zoomer.

1.2. Application : chute d'une bille dans le glycérol

Il s'agira encore d'une équation différentielle linéaire.

Une bille d'acier, de rayon R , est lâchée dans le glycérol. On indique qu'elle est soumise à une force de frottements fluide :

$$\vec{f} = -6\pi\eta R\vec{v}$$

On donne : la masse volumique de l'acier $\rho_1 = 7800 \text{ kg.m}^{-3}$, du glycérol $\rho_2 = 1260 \text{ kg.m}^{-3}$, l'intensité de la pesanteur $g = 9,81 \text{ m.s}^{-2}$ et la viscosité du glycérol $\eta = 1 \text{ Pa.s}$.

On orientera vers le haut l'axe vertical ; la bille touche le glycérol à $t = t_0 = 0$ avec une vitesse initiale v_0 . On prendra $v_0 = 0$, dans un premier temps. (on notera v_0 la variable correspondante).

Exercice E2

1. Montrer que l'équation différentielle vérifiée par la vitesse de la bille peut se mettre sous la forme :

$$\frac{dv}{dt} = -\frac{9\eta}{2\rho_1 R^2} v + \left(\frac{\rho_2}{\rho_1} - 1 \right) g$$

2. Pour une bille de diamètre $R = 2 \text{ mm}$, utiliser la fonction `ode` pour renvoyer un vecteur ligne v_1 correspondant à $v(t)$ pour 1000 valeurs de t entre 0 et 0,1 s (vecteur-ligne à 1000 colonnes).

3. Tracer v_1 en fonction de t .

4. Pour une bille de diamètre $R = 3 \text{ mm}$, utiliser la fonction `ode` pour renvoyer un vecteur ligne v_2 correspondant à $v(t)$ pour 1000 valeurs de t entre 0 et 0,1 s.

5. Tracer v_2 en fonction de t . On superposera les deux courbes sur un même graphique.

On veut maintenant observer l'influence de la condition initiale v_0 sur l'allure de la solution v_1 .

6. Écrire une boucle permettant de tracer sur un même graphique chaque solution v_1 correspondant à des valeurs de v_0 différentes allant de $-0,12$ à 0 avec un pas de $0,02$.

2. ÉQUATIONS DIFFÉRENTIELLES D'ORDRE 2

2.1. Transformation d'une équa. diff. d'ordre 2 en équa. diff. vectorielle d'ordre 1

Méthode illustrée sur une équation différentielle linéaire.

Considérons une équa. diff. du second ordre, $y(t)$ étant une fonction scalaire : $\frac{d^2y}{dt^2} + b(t)\frac{dy}{dt} + a(t)y(t) = c(t)$.

On peut la réécrire sous la forme : $y''(t) = c(t) - b(t)y'(t) - a(t)y(t)$.

On va la ramener à une équa. diff. du premier ordre mais dont l'inconnue sera une fonction vectorielle.

Intérêt : on pourra lui appliquer alors la commande `ode` qui, suivant la philosophie de Scilab, s'applique aussi à des fonctions vectorielles.

Idée : on pose $\bar{Y}(t)$ ou $\mathbf{Y}(t)$ $\left| \begin{array}{l} Y_1 = y(t) \\ Y_2 = y'(t) \end{array} \right.$

Les deux composantes n'ayant pas la même dimension, la flèche n'a pas vraiment de sens. Nous noterons donc ces vecteurs en caractères gras.

En dérivant, on obtient $\mathbf{Y}'(t)$ $\left| \begin{array}{l} Y'_1 = y'(t) = Y_2 \\ Y'_2 = y''(t) = c(t) - b(t)y'(t) - a(t)y(t) = c(t) - b(t)Y_2 - a(t)Y_1 \end{array} \right.$

Notre équa. diff. peut donc s'écrire comme une équation du premier ordre pour la fonction $\mathbf{Y}(t)$ sous la forme :

$\mathbf{Y}'(t) = f(t, \mathbf{Y}(t))$ avec $f : (t, \mathbf{Y}(t)) \mapsto f(t, \mathbf{Y}(t)) = \left| \begin{array}{l} Y_2 \\ c(t) - a(t)Y_1 - b(t)Y_2 \end{array} \right.$

2.2. Traduction en Scilab : exemple de l'oscillateur harmonique non amorti

On considère l'équa. diff. d'un oscillateur harmonique non amorti suivant Oy : $y'' + \omega_0^2 y = 0 \Leftrightarrow y''(t) = -\omega_0^2 y(t)$

On introduit alors $\mathbf{Y}(t)$ $\left| \begin{array}{l} Y_1 = y(t) \\ Y_2 = y'(t) \end{array} \right.$ pour avoir l'équa. diff. sous la forme : $\mathbf{Y}'(t) = f(t, \mathbf{Y}(t))$ avec $f(t, \mathbf{Y}(t)) = \left| \begin{array}{l} Y_2 \\ -\omega_0^2 Y_1 \end{array} \right.$

Application en Scilab : on se donne la pulsation propre $\omega_0 = 2 \text{ rad.s}^{-1}$, la position initiale $y(0) = 1 \text{ cm}$, la vitesse initiale $y'(0) = 0$. On a donc $\mathbf{Y}(0) = \left| \begin{array}{l} 1 \\ 0 \end{array} \right.$. On veut tracer la solution, c'est-à-dire le graphe de l'équation horaire $y(t)$.

On peut utiliser le script suivant :

```

w = 2; // pulsation propre
function Yprime = Yprime(t, Y)
  Yprime(1)=Y(2) // première composante de Yprime (y') )= 2e composante de Y
  Yprime(2)=-w^2*Y(1) // 2e composante de Yprime = y''
endfunction
Y0 = [1;0] // C.I. rentrées comme un vecteur vertical
//Y0=[1 0]' //alternative
//y0=1; v0=0; Y0=[y0;v0] //ou encore
t0 = 0
t = linspace(0,10,1000)
Y = ode(Y0,t0,t,Yprime) // Y est un tableau à 2 lignes et length(t) colonnes
plot2d(t,Y(1,:),style=2) // on ne trace que la 1e ligne de Y, soit y(t)

```

2.3. Rappels importants, utiles pour bien comprendre l'exemple précédent

Attention : un vecteur-ligne n'est pas équivalent à un vecteur-colonne

a) fabrication d'un vecteur composantes après composantes

Par défaut si un vecteur u n'existe pas et qu'on rentre dans la console :

```

-->u(1)=2;
-->u(2)=9
u =
    2.
    9.

```

On a fabriqué un vecteur à deux lignes et une colonne, autrement dit un vecteur-colonne à 2 composantes.

b) application à la fonction `Yprime` définie au §2.2

Cette fonction, pour chaque valeur de t , prend en entrée un vecteur Y qui peut indifféremment être un vecteur-ligne ou un vecteur-colonne, car les commandes `Y(1)` et `Y(2)` s'appliquent dans les deux cas. En revanche elle renvoie un vecteur-colonne, puisque les composantes sont calculées les unes à la suite des autres, comme expliqué en a).

La variable de retour de `ode` est un tableau à deux lignes et à autant de colonnes qu'il y a de valeurs de t , avec en première ligne les valeurs de $y(t)$ et en seconde ligne les valeurs de $y'(t)$.

c) commande d'extraction dans un tableau

Retenir en particulier la commande d'extraction d'une ligne (ou d'une colonne) dans un tableau :

```

-->T=[1 2 3
-->4 5 6]
T =
    1.    2.    3.
    4.    5.    6.
-->T(1)
ans =
    1.
-->T(1,2)
ans =
    2.
-->T(2,1)
ans =
    4.
-->T(1,:)
ans =
    1.    2.    3.
-->T(2,:)
ans =
    4.    5.    6.
-->T(:,1)
ans =
    1.
    4.

```

2.4. Application au portrait de phase de l'oscillateur harmonique non amorti

a) énergie mécanique

Avec un axe Oy horizontal convenablement placé (origine coïncidant avec la position d'équilibre) et une origine de l'énergie potentielle en $y = 0$, l'énergie mécanique d'un oscillateur harmonique s'exprime par : $\mathcal{E}_m = \mathcal{E}_c + \mathcal{E}_p = \frac{1}{2}m(y')^2 + \frac{1}{2}ky^2$

Exercice E3

Retrouver cette expression à partir du problème du mouvement horizontal sans frottements d'une masse m accrochée à un ressort de raideur k .

b) une courbe dans l'espace des phases

Par définition, la courbe *dans l'espace des phases* correspondant à une solution de l'équation de l'oscillateur harmonique est la courbe paramétrée $t \mapsto (y(t), y'(t))$. Avec le résultat du §2.2, son tracé est immédiat :

```
scf(1);
plot2d(Y(1,:), Y(2,:))
```

On obtient bien des ellipses comme prévus par l'équation définissant l'énergie mécanique. Mieux, si on prend comme variables $\left(y(t), \frac{y'(t)}{\omega_0}\right)$ on obtient un cercle puisque : $\frac{1}{2}m(y')^2 + \frac{1}{2}ky^2 = \mathcal{E}_m \Leftrightarrow y^2 + \left(\frac{y'}{\omega_0}\right)^2 = \frac{2\mathcal{E}_m}{k} = cste$

ce qui se voit bien à condition de prendre des coordonnées en base orthonormée :

```
scf(2);
plot2d(Y(1,:), Y(2,:)/w)
A=gca(); // accès aux propriétés des axes
A.isoview="on"; // base orthonormée
```

Exercice E4 : tout un portrait

Pour avoir tout un portrait de phase, on doit faire varier les conditions initiales : comment obtenir en Scilab un tracé simultané pour différentes conditions initiales ?

On pourra par exemple garder une vitesse initiale nulle et faire varier l'amplitude initiale.

2.5. Oscillations d'un pendule simple non amortia) équation différentielle du mouvement**Exercice E5**

Montrer qu'en l'absence de frottements, l'équation différentielle du mouvement d'un pendule simple de longueur ℓ est : $\ddot{\theta} + \omega_0^2 \sin \theta = 0$ avec $\omega_0 = \sqrt{\frac{g}{\ell}}$

b) graphe du mouvement

Nous ne savons pas résoudre cette équation dans le cas général, car elle n'est pas linéaire. Seul le cas des oscillations de faible amplitude est à notre portée : c'est le cas où $\sin \theta \approx \theta$, qui conduit alors à l'équation différentielle d'un oscillateur harmonique.

Par contre, Scilab peut nous donner le graphe du mouvement dans tous les cas de figure.

Exercice E6


On lâche le pendule sans vitesse initiale, à partir d'une position initiale θ_0 .

Pour différentes valeurs de θ_0 , comprises entre 0 et 1 rad, tracer la courbe $\theta(t)$, pour $t \in [0; 10\text{s}]$.

Dans Scilab, on pourra remplacer θ par α par exemple.

Comparer aux courbes obtenues pour un oscillateur harmonique, avec les mêmes conditions.

On prendra $g = 9,81 \text{ m}\cdot\text{s}^{-2}$ et $\ell = 0,50 \text{ m}$.



Source principale :

<https://joffrempsi1.files.wordpress.com/2013/09/s2-edo.pdf>